# $Id: PKGBUILD 83136 2013-01-27 16:13:00Z pierre $
# Maintainer: Aaron Griffin <aaron@archlinux.org>
# Contributor: Travis Willard <travisw@wmpub.ca> 
# Contributor: William Rea <sillywilly@gmail.com>

pkgname=devilspie
pkgver=0.22
pkgrel=5
pkgdesc="x11 window matching utility"
arch=('i686' 'x86_64')
url="http://www.burtonini.com/blog/computers/devilspie"
license=('GPL')
depends=('libwnck')
makedepends=('pkgconfig' 'intltool' 'gnome-common')
source=("http://www.burtonini.com/computing/$pkgname-$pkgver.tar.gz")
md5sums=('4190e12f99ab92c0427e457d9fbfe231')

build() {
  cd $srcdir/$pkgname-$pkgver
  sed -i -e '/-DG.*_DISABLE_DEPRECATED/d' src/Makefile.am
  export LIBS="-lX11"
  ./autogen.sh

  ./configure --prefix=/usr --sysconfdir=/etc
  make
}

package() {
  cd $srcdir/$pkgname-$pkgver
  make prefix=$pkgdir/usr install
  # Make and install blank config
  mkdir -p $pkgdir/etc/devilspie
  echo '(debug)' > $pkgdir/etc/devilspie/example.ds
}
